## ANNUAL RETENTION PRODUCT ANALYSIS REPORT OVERVIEW

This Power BI report meticulously examines customer retention and sales performance across product categories, from one year to the next. Essential for tracking the efficacy of marketing initiatives and outreach programs, this analysis checks whether customers who made purchases one year continue to do so the next.

Leveraging advanced DAX Measures, the analysis provides precise, dynamic calculations that mirror critical performance indicators, thus enabling stakeholders to gauge sales success, track emerging trends, and pinpoint potential areas for growth. These measures are pivotal, offering immediate clarity on retention metrics, including tracking customers who placed orders in consecutive years, those who didn't, and analyzing year-over-year churn and retention rates.

The bespoke visuals not only elevate the report's look but also streamline understanding for diverse audiences. This exhaustive analysis empowers organizations to steer data-backed decisions, fine-tune strategies, and bolster business results.

This comprehensive analysis equips organizations to pursue informed, data-driven strategies, refine tactics, and optimize business performance. Highlighting everything from the sophisticated data models to the expertly crafted DAX Measures and visually stunning custom backgrounds, this Power BI solution demonstrates our profound commitment to excellence in analytics and strategic decision-making.




| Category | Y1 \| Profit | Y2 \| Profit | Y1 \| Orders | Y2 \| Orders | Y1 \| Customers | Y2 \| Customers | Repeat Customers |
|---|---|---|---|---|---|---|---|
| Furniture | $3,012.00 | $3,788.00 | 452 | 295 | 289 | 128 | 128 |
| Office Supplies | $25,101.00 | $26,079.00 | 1241 | 964 | 502 | 337 | 337 |
| Technology | $33,493.00 | $21,520.00 | 409 | 266 | 276 | 96 | 96 |
| **Total** | **$61,606.00** | **$51,387.00** | **2102** | **1525** | **573** | **452** | **452** |

## DATA MODEL OVERVIEW

Model Structure:

- Star schema

Fact Table:

- Fact Order Table

Dimension Table(s):

- Dim Product Category
- Dim Product Sub Category
- Dim Customer Table
- Dim Calendar

Disconnected Table(s):

- Sales Measures
- Annual Retention Measures
- Year 1 & 2 Tables



## ANNUAL PRODUCT RETENTION REPORT

Visual Elements:

- Retention Trend Line: A graph illustrating the retention of customers in both selected years. This helps in visualizing the peaks and troughs of customer retention trends.
    - This visual can easily be adapted to include additional insights such as the count of customers who didn't order in the second year and the products they did or didn't order.
- Comparison Table: Tabular data showing the detailed comparison of Year 1 and Year 2 metrics. These metrics include:
    - Year 1 and Year 2 Profits
    - Year 1 and Year 2 Orders
    - Year 1 and Year 2 Customers
    - Repeat Customers

KPI Cards:

- Year 1 and Year 2 Profits: A comparison of profits between the two years, which can be displayed in a card format for quick reference.
- YoY Delta: The change in profit from Year 1 to Year 2.
- YoY Change: The percentage change in profit from Year 1 to Year 2.
- YoY Churn and Retention Rates: Two separate KPI cards showing the percentages of customer churn and retention respectively.

Navigation Bars and Filters:

- Product Category Navigation: Tabs or buttons that allow users to filter the displayed data by product category.
- Year Selector: Buttons or dropdown menus that enable users to choose between Year 1 and Year 2 for comparative analysis.

## DAX MEASURES

Below are some of the custom Data Analysis Expressions (DAX) measures that are driving the reports.

```
Year 1 Customers =
VAR SelectedYear = SELECTEDVALUE('Year 1'[Year 1])
RETURN
    IF(ISBLANK(SelectedYear),        0,
        CALCULATE(
            DISTINCTCOUNT('Fact Order Table'[Customer ID]),
            'Fact Order Table'[Order Year] = SelectedYear,
           'Fact Order Table'[Sales Revenue] > 0
        )
    )
```

```
Year 2 Customers =
VAR SelectedYear1 = SELECTEDVALUE('Year 1'[Year 1])  -- Year 1 selection
VAR SelectedYear2 = SELECTEDVALUE('Year 2'[Year 2])  -- Year 2 selection
VAR CustomersWithPositiveGrossYear1 =
    CALCULATETABLE(
        VALUES('Fact Order Table'[Customer ID]),
        'Fact Order Table'[Order Year] = SelectedYear1,
        'Fact Order Table'[Sales Revenue] > 0    )
VAR CustomersWithPositiveGrossYear2 =
    CALCULATETABLE(
        VALUES('Fact Order Table'[Customer ID]),
        'Fact Order Table'[Order Year] = SelectedYear2,
        'Fact Order Table'[Sales Revenue] > 0    )
VAR CommonCustomersWithPositiveGrossInBothYears =
    INTERSECT(CustomersWithPositiveGrossYear1, CustomersWithPositiveGrossYear2)
RETURN
    IF(
        SelectedYear2 <= SelectedYear1,  -- Ensure Year 2 is after Year 1
        BLANK(),  -- Logical check to ensure comparison makes sense
        COALESCE(
            COUNTROWS(CommonCustomersWithPositiveGrossInBothYears),0))
```

```
YoY Churn =
IF(ISBLANK([Customers With Positive Sales In Both Years]),
    BLANK(),
    1 - [YoY Retention])
```

```
YoY Retention =
IF(
    ISBLANK([Customers With Positive Sales In Both Years]) || ISBLANK([Year 1 Customers]),
    BLANK(),
    DIVIDE([Customers With Positive Sales In Both Years], [Year 1 Customers]))
```

```
Count Customers With No Sales In Year 2 =
VAR SelectedYear1 = SELECTEDVALUE('Year 1'[Year 1])
VAR SelectedYear2 = SELECTEDVALUE('Year 2'[Year 2])
VAR CustomersInYear1 =
    CALCULATE(
        DISTINCTCOUNT('Fact Order Table'[Customer ID]),
        'Fact Order Table'[Order Year] = SelectedYear1,
        'Fact Order Table'[Sales Revenue] > 0     )
VAR CustomersWithSalesInBothYears =
    [Customers With Positive Sales In Both Years]
VAR CustomersNoSalesInYear2 =
    IF(
        SelectedYear2 <= SelectedYear1 || ISBLANK(CustomersWithSalesInBothYears),
        BLANK(),  -- Return BLANK if Year2 is not after Year1 or CustomersWithSalesInBothYears is BLANK
        IF(
            CustomersInYear1 - CustomersWithSalesInBothYears = 0,
            BLANK(),  -- Additionally, return BLANK if the result is 0, indicating no difference in counts
            CustomersInYear1 - CustomersWithSalesInBothYears
        ))
RETURN
    CustomersNoSalesInYear2
```

```
Customers With Positive Sales In Both Years =
VAR SelectedYear1 = SELECTEDVALUE('Year 1'[Year 1])
VAR SelectedYear2 = SELECTEDVALUE('Year 2'[Year 2])
VAR CurrentYear = YEAR(TODAY())  -- Get the current year
VAR MaxYearInData = MAX('Fact Order Table'[Order Year])  -- Or use the maximum year in your data if working with
future data
VAR CustomersWithPositiveSalesInYear1 =
    CALCULATETABLE(
        VALUES('Fact Order Table'[Customer ID]),
        'Fact Order Table'[Order Year] = SelectedYear1,
        'Fact Order Table'[Sales Revenue] > 0
    )
VAR CustomersWithPositiveSalesInYear2 =
    CALCULATETABLE(
        VALUES('Fact Order Table'[Customer ID]),
        'Fact Order Table'[Order Year] = SelectedYear2,
        'Fact Order Table'[Sales Revenue] > 0
    )
VAR CommonCustomersWithPositiveSales =
    INTERSECT(CustomersWithPositiveSalesInYear1, CustomersWithPositiveSalesInYear2)
RETURN
    IF(
        SelectedYear2 <= SelectedYear1 || SelectedYear2 > CurrentYear || SelectedYear2 > MaxYearInData,
        BLANK(),  -- Return 0 if Year2 is not valid or in the future
        COUNTROWS(CommonCustomersWithPositiveSales))
```

```
YoY Profit Delta =
VAR SelectedYear1Value = SELECTEDVALUE('Year 1'[Year 1])
VAR SelectedYear2Value = SELECTEDVALUE('Year 2'[Year 2])
VAR SelectedYear1Profit = [Year 1 Profit]
VAR SelectedYear2Profit = [Year 2 Profit]
RETURN
    IF(
        SelectedYear2Value <= SelectedYear1Value,  -- Checks if year_2 is the same as or before year_1
        0,  -- Return 0 if true
        COALESCE(
            IF (
                NOT ISBLANK(SelectedYear2Profit) &&
                NOT ISBLANK(SelectedYear1Profit),
                SelectedYear2Profit - SelectedYear1Profit
            ),
            0
        )
    )
```